Marder 6.8.

```
> restart;
```

Left hand side of equation from earlier work:

```
> LHS:=hbar^3*Pi^2*n/(m*sqrt(2*m)*(k*T)^(3/2));
```

$$LHS := \frac{hbar^3 \pi^2 n \sqrt{2}}{2\, m^{(3/2)} (k\, T)^{(3/2)}}$$

Right hand side -- express the integral as a function of chemical potentail over kT.

```
> f:=betamu->evalf(int(sqrt(x)/(exp(x-betamu)+1),x=0..10000));
```

$$f := betamu \rightarrow evalf\left(\int_0^{10000} \frac{\sqrt{x}}{\mathbf{e}^{(x - betamu)} + 1}\, dx\right)$$

```
> hbar:=1.054E-34;m:=9.11E-31;T:=300;n:=18.1E28;k:=1.381E-23;
```

$$hbar := 1.054\ 10^{-34}$$

$$m := 9.11\ 10^{-31}$$

$$T := 300$$

$$n := 1.81\ 10^{29}$$

$$k := 1.381\ 10^{-23}$$

```
> evalf(LHS);
```

$$6378.744250$$

We need to do the integral with different values of the mu/kT until we find an integral that yields the correct value. Fortunately, Maple's fsolve function will do that for us.

```
> muOverKT:=fsolve(f(y)=LHS,y);
```

$$muOverKT := 450.6947932$$

```
> mu:=k*T*muOverKT/1.602E-19;
```

$$\mu := 11.65560879$$

The above is the value of the chemical potential at 300K.

Now let's calculate mu for 10,000K:

```
> T:=10000;
```

$$T := 10000$$

```
> muOverKT:=fsolve(f(y)=LHS,y);
```

$$muOverKT := 13.45955291$$

```
> mu:=k*T*muOverKT/1.602E-19;
```

$$\mu := 11.60277314$$

At T=0, the chemical potential is equal to the fermi energy Ef. (See equations 6.29 and 6.31.)

```
> Ef:=hbar^2*kf^2/(2*m);
```

$$Ef := 6.097233810\ 10^{-39}\ kf^2$$

```
> kf:=(3*Pi^2*n)^(1/3);
```

$$kf := 5.656652826\ 10^{9}\ 3^{(1/3)}\ (\pi^2)^{(1/3)}$$

```
> Ef_eV:=evalf(Ef/1.602E-19);
```

$$Ef\_eV := 11.65565599$$

So, the chemical potential is essentially unchanged between 0 and 300K. At 10,000K, the potential decreases by

```
> delta_mu:=Ef_eV-mu;
```

$$delta\_mu := 0.05288285$$

According to the Sommerfeld expansion, the decrease is

```
> delta_mu_sommerfeld:=evalf((Pi^2/6)*(k*T)^2/(2*Ef))/1.6E-19;
```

$$delta\_mu\_sommerfeld := 0.05250330488$$

These differ by a fraction

```
> (delta_mu-delta_mu_sommerfeld)/delta_mu_sommerfeld;
```

$$0.007228975792$$

or a little less than 1%.